
SECONDO TUTORIAL SU PSPICE5 (DOS) 
 
Insieme al file PDF del tutorial c’è la cartella con i file “*.CIR” dei circuiti di esempio. 
Questo tutorial non è la continuazione del primo tutorial. 
 
Qualcosa su SPICE e PSPICE  
 
Spice (Simulation Program with Integrated Circuit Emphasis) è un software sviluppato nel 1970 nell’Università di 
Berkeley per la simulazione di circuiti elettronici, come strumento di analisi, di progetto e prova di circuiti integrati: ma 
fornisce ottimi risultati anche nei circuiti elettrici . Spice è un software “public domain”: la versione commerciale, come 
PSpice della MicroSim Corporation, sfrutta lo stesso algoritmo e sintassi, fornendo risultati con la medesima precisione. 
In questo tutorial si introducono gli elementi basilari di Spice/PSpice per le applicazioni su semplici circuiti elettrici. 
 
 
Descrizione dei circuiti (NETLIST)  
 
La NETLIST è scritta come una sequenza di righe contenute in un file di testo creato da un editor di testo ASCII e 
chiamato FILE SORGENTE: esso può essere creato graficamente costruendo il circuito con lo SCHEMATIC del 
programma commerciale della Microsim (esistono versioni gratuite di PSPICE come il Pspice 9.1 e il Pspice 8.0, che si 
trovano facilmente sui siti universitari o scolastici). 
In questo tutorial, come già fatto nel precedente tutorial, i file sorgente saranno dei file “*.CIR ”: per analizzare il circuito 
si esegue (da una finestra DOS, nella cartella del PSPICE5 scompattato), il file sorgente, ottenendo l’omonimo file 
OUTPUT “*.OUT ”, che contiene i risultati della simulazione circuitale oppure i messaggi di errore che il software ha 
rilevato (errori di sintassi, o altro). 
 
 
Riga del TITOLO  
 
È obbligatorio inserire la prima riga nel file sorgente (anche se si tratta di una riga vuota): essa non serve nell’analisi e 
PSPICE la riporta tale e quale nel file di OUTPUT. Tale riga può essere usata per dare un titolo al circuito e non 
necessita di nessun carattere iniziale, per esempio l’asterisco o altro. 
 
 
Riga FINALE “.END”  
 
La riga finale del file sorgente è costituita da “.END”: qualsiasi riga scritta dopo tale riga sarà elaborata da PSPICE come 
un nuovo file sorgente. 
 
 
Righe dei dati  
 
Le righe intermedie nel file sorgente specificano completamente il circuito e il tipo di analisi desiderati. Nell’Esempio 1 la 
seconda riga descrive un generatore di tensione chiamato Vs collegato tra i nodi 1 (morsetto positivo) e il NODO DI 
RIFERIMENTO 0 (morsetto negativo): è un generatore in corrente continua (DC) con valore pari a 18 V. La terza riga 
descrive un resistore chiamato R1 collegato tra i nodi 1 e 2 con valore pari a 3k Ohm. Analogamente le righe quarta e 
quinta definiscono rispettivamente il resistore R2 (6k Ohm), e il condensatore C (5 mF), entrambi collegati tra i nodi 0 e 2. 
PSPICE richiede necessariamente il nodo 0 usato come riferimento(ground). 
Le righe dei dati descrivono la topologia del circuito (cioè quali componenti ci sono e come sono collegati fra loro): la 
sintassi è descritta brevemente più avanti e in maniera più puntuale nel file “DEVICE.PDF”. 
 
 
Righe di Controllo e di Uscita  
 
In assenza di ulteriori comandi alle righe dei dati, PSPICE calcola automaticamente il punto di lavoro in corrente continua 
(“dc steady state”, quindi un regime stazionario), delle seguenti variabili: 

(i) tensioni nodali rispetto al nodo di riferimento 0. 
(ii) le correnti entranti in ogni generatore di tensione. 
(iii) la potenza totale dissipata del circuito. 

 
Comunque possono essere aggiunte righe di comando nel file sorgente per specificare altre variabili di uscita o altri tipi 
di analisi: la sintassi è descritta brevemente più avanti e in maniera più puntuale nel file “COMMAND.PDF”. 
 
 
Righe dei dati e analisi in corrente continua (DC)  
Bipoli Passivi  
 
Le righe dei dati per I bipoli passivi R,L e C è formata da un minimo di tre elementi: il primo elemento fornisce il nome 
come una stringa di caratteri che inizia con R,L o C rispettivamente per resistori, induttori e condensatori; il secondo 
elemento specifica i nodi separati da uno spazio; il terzo elemento fornisce il valore in Ohm, Henry e Farad (meglio non 
specificare l’unità di misura, soprattutto i Farad!). 
Riguardo al valore dei bipoli si possono usare dei suffissi (come specificato in DEVICE.PDF), ma è più facile usare la 
notazione scientifica (p.e. 5 nF = 5E-9; 10 kOhm = 10E3; 0,0015 H = 1.5E-3,ecc.). 

Generatori Indipendenti  
I Generatori Indipendenti sono specificati dalla seguente sintassi: 
 

<name> <nodes> <type> <value>  
 
<name> = V oppure I rispettivamente per generatori di tensione o di corrente. 
<node > = sono i nodi di collegamento: per il generatore di tensione il primo nodo indica il terminale positivo, mentre per 

il generatore di corrente, la corrente scorre dal primo al secondo nodo, cioè viene erogata dal secondo nodo. 
<type > = può essere DC oppure AC rispettivamente per generatori in corrente continua o alternata. 
 
 
Generatori Controllati o Dipendenti  
I Generatori Linearmente Dipendenti sono specificati dalla seguente sintassi: 
 

<name> <nodes> <control> <gain>  
 
Ogni generatore è identificato dalla lettera iniziale che potrà essere: 

• lettera E per i generatori di tensione controllati in tensione (Exx); 
• lettera F per i generatori di corrente controllati in corrente (Fxx); 
• lettera G per i generatori di corrente controllati in tensione (Gxx); 
• lettera H per i generatori di tensione controllati in corrente (Hxx). 

 
L’ordine dei nodi è lo stesso dei generatori indipendenti corrispondenti: per i generatori controllati in tensione (Exx e 
Gxx), il “controllo” (control), è la coppia di nodi (prima il positivo e poi il negativo), della differenza di potenziale da cui 
dipende il generatore; per i generatori controllati in corrente (Fxx e Hxx), il “controllo” è una corrente in un ramo (a volte è 
necessario mettere un generatore indipendente di tensione di valore nullo che quindi viene usato come amperometro). 
Il “guadagno” “gain ”, è il fattore di proporzionalità specifico del generatore controllato. 
 
ESEMPIO 1: calcolare con PSPICE la tensione in regime stazionario ai capi del condensatore. 

−

+18V

3kΩ

6kΩ 5µF
−

+Vs 18V

1 R1

3kΩ

2

R26kΩ

0

C5µF

(a) (b)  
 
Prima operazione consiste nel mettere i nodi partendo da 0 e dare dei nomi a tutti i bipoli R1, R2, C, e Vs. 
Si crea quindi il file sorgente ASCII con le righe riportate di seguito denominato ESE_01.CIR  
 
ANALISI DC – ESEMPIO 1 
Vs 1 0 DC 18 
R1 1 2 3E3 
R2 0 2 6E3 
C 0 2 5E-6 
.END 
 
Il file sorgente si trova nella stessa cartella PSPICE5 dove eseguire il comando (finestra DOS): 
 

PSPICE ESE_01 
 
Verrà creato il file di uscita ESE_01.OUT con i risultati della simulazione: come detto prima, in assenza di ulteriori 
comandi alle righe dei dati, PSPICE calcola automaticamente il punto di lavoro in corrente continua (dc steady-state): 
 
NODE VOLTAGE NODE VOLTAGE 
(1) 18.0000  (2) 12.0000 
VOLTAGE SOURCE CURRENTS 
NAME  CURRENT 
Vs  -2.000E-03 
TOTAL POWER DISSIPATION 3.60E-02  WATTS 
 
Il file di uscita specifica che la tensione nodale V(2), cioè prendendo come riferimento il nodo 0, vale 6V, mentre la 
corrente ENTRANTE nel generatore di tensione è - 2mA ( negativa perché la corrente è positiva e uscente); infine la 
potenza totale dissipata nel circuito è di 36 mW (cioè la potenza dissipata dalle due resistenze). 

Ing.Carmelo Gerardi

ESEMPIO 2: effettuare l’analisi in regime stazionario con PSpice. 
1.

500Ω

3kΩ 1kΩ

1.5kΩ

−+

5V

4mA R1 500Ω

1
R2

3kΩ

2 R3

1kΩ
3

R41.5kΩ

−+

Vs

5V

0

Is 4mA

(a) (b)  
 

Si passa quindi alla figura (b) numerando i nodi e nominando i bipoli nel file sorgente (ESE_02.CIR ), scritto di seguito: 
 

ANALISI DC – ESEMPIO 2 
R1 0 1 500 
R2 1 2 3E3 
R3 2 3 1E3 
R4 0 3 1.5E3 
Vs 3 1 DC 5 
Is 0 2 DC 4E-3 
.END 
 

Nel file di uscita che la simulazione con Pspice produce, si possono leggere le tensioni nodali e la corrente erogata 
(negativa) dal generatore di tensione, in regime stazionario: 
 

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE 
(1) .2500  (2) 7.0000  (3) 5.2500 
VOLTAGE SOURCE CURRENTS 
NAME  CURRENT 
Vs  -1.750E-03 
TOTAL POWER DISSIPATION 8.75E-03 WATTS 
 
 
ESEMPIO 3 effettuare l’analisi in regime stazionario con PSpice. 

1.

−

+ 20V

1kΩ

R22kΩ 500Ω

75 · ii

−

+Vs 20V

1
R1

1kΩ

2 3
−+

V amp

0V

R3 500Ω

4

Is

75 · i

3

R22kΩ

0

i

(a) (b)  
 

Si passa quindi alla figura (b) numerando i nodi e nominando i bipoli nel file sorgente (ESE_03.CIR ), scritto di seguito. 
Si è inserito un generatore Vdmy (di valore nullo), che fornisce la corrente di “controllo” del generatore dipendente. 
 

ANALISI DC CON GENERATORI DIPENDENTE - ESEMPIO 3 
Vs 1 0 DC 20 
R1 1 2 1E3 
R2 0 3 2E3 
R3 0 4 500 
Vamp 2 3 0 
FIs 4 3 Vamp 75 
.END 
 

Nel file di uscita che la simulazione con Pspice produce, si possono leggere le tensioni nodali e la corrente erogata 
(negativa), dal generatore di tensione, in regime stazionario: 
 

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE 
(1) 20.0000 (2) 19.8690 (3) 19.8690 (4) -4.9020 
VOLTAGE SOURCE CURRENTS 
NAME  CURRENT 
Vs  -1.307E-04 
Vamp          1.307E-04 
TOTAL POWER DISSIPATION 2.61E-03 WATTS 

Ing.Carmelo Gerardi



4 

ESEMPIO 4: Nel circuito seguente calcolare il valore di Vs tale che sia nulla la potenza dissipata nella resistenza R1. 
Usare il comando “.DC” per variare il valore di Vs da 1 a 6 V, e anche il comando “.PRINT” per visualizzare la corrente 
del generatore I(Vs), e le tensioni su R1 cioè V(1,2), e su R2 cioè V(2). 

−

+2V

1kΩ

2kΩ 2mA
−

+Vs 2V

1
R1

1kΩ

2

R22kΩ

0

Is 2mA

(a) (b)  
Il file sorgente (ESE_04.CIR ) è scritto di seguito: 
 

ANALISI DC (SWEEP) - ESEMPIO 4 
Vs 1 0 DC 2 
Is 0 2 DC 2E-3 
R1 1 2 1E3 
R2 0 2 2E3 
.DC Vs 2 6 1 
.PRINT DC I(Vs) V(1,2) V(2) 
.END 
 

Nel file di uscita che la simulazione con Pspice produce, si possono leggere la corrente e le tensioni richieste: 
 

DC TRANSFER CURVES 
Vs   I(Vs)   V(1,2)   V(2) 
2.000E+00     6.667E-04    -6.667E-01     2.667E+00  
3.000E+00     3.333E-04    -3.333E-01     3.333E+00  
4.000E+00    -2.667E-12     2.667E-09     4.000E+00  
5.000E+00    -3.333E-04     3.333E-01     4.667E+00  
6.000E+00    -6.667E-04     6.667E-01     5.333E+00  
 

Il valore di Vs che annulla la potenza dissipata nella resistenza R1 è Vs = 4 V. 
 
 
Comando “ .TF”: Equivalente di THEVENIN 
La riga di commando “.TF” fornisce la funzione di trasferimento, cioè si studia il rapporto tra una tensione di uscita 
variabile rispetto a una tensione di ingresso anch’essa variabile. Tale comando fornisce anche la resistenza vista dalle 
due tensioni, cioè la resistenza di uscita e di ingresso: dai risultati del comando “.TF” si possono ricavare la resistenza e 
la tensione di Thevenin. La sintassi del comando è: 
 

.TF  <output variable>  <input variable> 
 

ESEMPIO 5 Determinare il circuito equivalente di Thevenin tra i morsetti A e B, usando il comando .TF 

−

+Vs 15V

1
R1

1kΩ

2

R2

2kΩ

+

V2

−

0

R3

1kΩ

3
A

B

−

+
E5 5 · V2

4

R4 200Ω

 
Il file sorgente (ESE_05.CIR ) è scritto di seguito: 
 

ANALISI DC ( TRANSFER FUNCTION) - ESEMPIO 5  
Vs 1 0 DC 15 
E5 4 0 2 0 5 
R1 1 2 1E3 
R2 2 0 2E3 
R3 2 3 1E3 
R4 3 4 200 
.TF V(3) Vs 
.END 
 

Nel file di uscita che la simulazione con Pspice produce quanto precedentemente detto: 
 

NODE  VOLTAGE  NODE  VOLTAGE  NODE  VOLTAGE  NODE  VOLTAGE 
(1)  15.0000  (2) -8.1818  (3) -35.4550  (4) -40.90 90 
VOLTAGE SOURCE CURRENTS 
NAME   CURRENT 
Vs  -2.318E-02 
TOTAL POWER DISSIPATION 3.48E-01 WATTS 
SMALL-SIGNAL CHARACTERISTICS 
V(3)/Vs = -2.364E+00 
INPUT RESISTANCE AT Vs = 6.471E+02 
OUTPUT RESISTANCE AT V(3) = -2.273E+02 
 

La tensione di Thevenin è ETh =[V(3)/Vs]*Vs=(-2,364)*(15)=-35,46 V = V(3), mentre la resistenza di Thevenin è la 
resistenza di uscita cioè RTh = - 227,3 ohm. 

Ing.Carmelo Gerardi

Regime Sinusoidale (AC) e Risposta in Frequenza  
Generatori Independenti AC  
Generatori Independenti AC sono descritti dalla riga di comando con la seguente sintassi: 

 
<name>  <nodes>  AC  <magnitude>  <phase in degrees> 

 
I campi <name> e <node > si scrivono come nei generatori DC, mentre <magnitude > e <phase > è il valore del fasore. 
 
ESEMPIO 6: Studiare la risposta in frequenza del circuito RLC serie seguente variando la frequenza del generatore da 
40 a 60 kHz in 200 “passi”. Determinare l’ampiezza e la fase della corrente I usando i comandi “.PLOT” e “.PROBE” 

−

+Vs cos(ωt)

1 R

32Ω

2
L

2mH

3

C5nF

00  
Il file sorgente (ESE_06.CIR ) è scritto di seguito: 
 

ANALISI AC - ESEMPIO 6  
Vs 1 0 AC 1 0 
R 1 2 32 
L 2 3 2E-3 
C 3 0 5E-9 
.AC LIN 200 40E3 60E3 
.PLOT AC IM(Vs) IP(Vs) 
.PROBE VM(1, 2) VM(2,3) VM(3) IM(Vs) IP(Vs) 
.END 
 
L’ampiezza e la fase della corrente si determinano con “IM(Vs)” e “IP(Vs)”; per ottenere la parte reale e immaginaria si 
usano “IR(Vs)” e “II(Vs)”.Il comando PROBE crea il grafico della risposta in frequenza come mostrato di seguito. 
 

 
 
MUTUA INDUTTANZA (induttori mutuamente accoppiati).  
La mutua induttanza tra due induttori è un altro componente il cui nome inizia con K: la sintassi del componente è: 
 

<name>  <inductor 1>  <inductor 2>  <coupling coeff icient> 
 

Il puntino NERO determina il segno del coefficiente di mutua induttanza K, per cui nella sintassi si scrivono le induttanze 
iniziando con il nodo col puntino come nell’esempio seguente, che descrive due induttanze mutuamente accoppiate. 
Per le due induttanze mutuamente accoppiate, il numero da inserire dopo i nodi, è il valore in Henry, mentre per il 
coefficiente di accoppiamento si inserisce il valore calcolato come mostrato dopo. 

2

L1

2H

1

3H

L2

4

3

M= 1.5H

 
 

Il coefficiente di accoppiamento si calcola così: K12 = M/ √ (L1 * L2) = 1,5/ √ (2 * 3) = 0,6124. Il file sorgente risulta: 
L1 1 2 2 
L2 3 4 3 
K12 L1 L2 0.6124 

Ing.Carmelo Gerardi

ESEMPIO 7 : Disegnare con la funzione “.PLOT” l’impedenza di ingresso Zin = V1/I1 nel circuito seguente variando la 
frequenza da 0.01 Hz a 1 Hz. 

0

Is1 6 0◦A

1

+
V1

−

C

1F

L1

2H

2

3Ω R

L2

5H

3

L31H

I1

Zin = V1 ÷ I1

M = 2H

 
 

Per ottenere la Zin, si inserisce tra i nodi 0 e 1 un generatore di corrente AC di 1 A e si disegna l’ampiezza e la fase della 
tensione V(1). Il coefficiente di accoppiamento si calcola così: K12 = M/ √ (L1 * L2) = 2/ √ (2 * 5) = 0,6325. 
Il file sorgente (ESE_07.CIR ) è scritto di seguito: 
 

ANALISI AC – Analisi di induttori mutuamente accopp iati - ESEMPIO 7 
Is 0 1 AC 1 0 
C 0 1 1 
R 0 2 3 
L1 1 2 2 
L2 3 2 5 
K12 L1 L2 0.6325 
L3 0 3 1 
.AC LIN 20 .01 1 
.PRINT AC Vm(1) Vp(1) 
.PROBE 
.END 
 

L’ampiezza e la fase della tensione V(1), cioè Vm(1) e Vp(1), che rappresentano ampiezza e la fase della Zin, sono 
disegnate usando Probe e il grafico è mostrato di seguito. Notare che il massimo si ha per una frequenza pari a 0,1 Hz. 

 
 
 
ESEMPIO 8: Usare PSpice per calcolare la tensione tra i nodi 3 e 4, V(3,4), nel circuito in figura. 

0

36ΩR1

1

12Ω R2

70Ω

R3

−

+ Vs100V

2 20Ω

R4

3

100Ω R5

4
30Ω

R6

+

V (3, 4)

−

 
 

Il file sorgente (ESE_08.CIR ) è scritto di seguito: 
 

ANALISI DC – ESEMPIO 8 
Vs 2 0 DC 100 
R1 0 1 36 
R2 0 1 12 
R3 1 2 70 
R4 2 3 20 
R5 3 4 100 
R6 4 0 30 
.DC Vs 110 110 1 
.PRINT DC V(1) V(3, 4) 
.END 
 

Ing.Carmelo Gerardi

Nel file di uscita si ottiene che: 
 
DC TRANSFER CURVES 
Vs   V(1)   V(3, 4) 
1.100E+02 1.253E+01 7.333E+01 
 

Perciò la tensione richiesta vale V(3,4)= 73,33 V. 
 
 
ESEMPIO 9: Determinare la corrente I4 nella resistenza R4 nel circuito in figura. 

0

18ΩR1

1

−

+ Vs150V

2

55ΩR2

6Ω

R3 3

Is15A 17ΩR4

I4

 
 

Il file sorgente (ESE_09.CIR ) è scritto di seguito: 
 

ANALISI DC – ESEMPIO 9 
VS 2 0 DC 150 
Is 0 3 DC 15 
R1 0 1 18 
R2 1 2 55 
R3 1 3 6 
R4 3 0 17 
.DC Vs 150 150 1 
.PRINT DC I(R4) 
.END 
 

Nel file di uscita si ricavano i seguenti risultati: 
 

DC TRANSFER CURVE 
Vs   I(R4) 
1.500E+02  9.037E+00 
 

La corrente richiesta è I(R4) e vale 9,037 A : si noti che il verso va dal nodo 3 al nodo 0 coerentemente all’ordine con cui 
si sono inseriti i nodi di collegamento della resistenza R4. 
 
 
ESEMPIO 10: Determinare le tre correnti di maglia nel circuito in figura usando PSpice. 

0

3ΩR1

1

−

+VS1 35V

2

8Ω R2

12Ω

R3 3

5ΩR4

3Ω

R5 4

−

+
VS255V

I1 I2 I3

 
 

Le tre correnti di anello coincidono con le correnti che circolano nelle resistenze con appartengono a una sola maglia 
(R1, R3 e R5): infatti le resistenze R2 e R4 sono entrambe attraversate da due correnti di maglia. 
 
Il file sorgente (ESE_10.CIR ) è scritto di seguito: 
 

ANALISI DC – ESEMPIO 10 
VS1 2 0 DC 35 
VS2 0 4 DC 55 
R1 0 1 3 
R2 1 2 8 
R3 1 3 12 
R4 3 0 5 
R5 3 4 3 
.DC V1 35 35 1 
.PRINT DC I(R1) I(R3) I(R5) 
.END 
 

Nel file di uscita si ricavano i seguenti risultati: 
 

DC TRANSFER CURVES 
V1   I(R1)   I(R3)   I(R5) 
3.500E+01 -1.193E+00  2.735E+00  8.585E+00 
 

In base all’ordine con cui si sono inserite le resistenze R1, R3 e R5 nel file sorgente, si deduce che le tre correnti di 
maglia avranno tutte e tre verso ORARIO. 

ESEMPIO 11 : mediante PSpice, calcolare il valore che deve avere Vs nell’ESEMPIO 10 tale che il generatore di 
tensione non eroghi nessuna potenza al circuito. Si usa la stessa sintassi dell’esempio 4 facendo variare Vs da 1 a 10V.  
 
Il file sorgente (ESE_11.CIR ) è scritto di seguito: 
 
ANALISI DC (SWEEP) - ESEMPIO 11 
R1 0 1 500 
R2 1 2 3E3 
R3 2 3 1E3 
R4 0 3 1.5E3 
Vs 3 1 DC 4 
Is 0 2 DC 3E-3 
.DC Vs 1 10 1 
.PRINT DC I(Vs) 
.PROBE 
.PLOT DC I(Vs) 
.END 
 
Il file di uscita contiene la “lista” dei risultati della simulazione: 
 
DC TRANSFER CURVES 
Vs   I(Vs) 
1.000E+00  7.500E-04 
2.000E+00 -2.188E-12 
3.000E+00 -7.500E-04 
4.000E+00 -1.500E-03 
5.000E+00 -2.250E-03 
6.000E+00 -3.000E-03 
7.000E+00 -3.750E-03 
8.000E+00 -4.500E-03 
9.000E+00 -5.250E-03 
1.000E+01 -6.000E-03 
 

Il generatore Vs non eroga potenza quando non eroga corrente cioè per Vs = 2V. 
 
 
ESEMPIO 12: Determinare l’ equivalente di Thevenin tra i terminali A - B, nel circuito in figura: 

0

−

+VS 6V

1

12Ω

R 2
A

+

IS 2A

−

 
 

Si inserisce il commando “.TF” come nell’esempio 5. Il file sorgente (ESE_12.CIR ) è scritto di seguito: 
 
ANALISI DC (Equivalente di Thevenin) - ESEMPIO 12 
Vs 1 0 DC 6 
R1 1 2 12 
Is 0 2 DC 2 
.TF V(2) Is 
.END 
 

Il file di uscita fornisce i risultati della simulazione: 
NODE   VOLTAGE   NODE   VOLTAGE 
(1)   6.0000   (2)   30.000 
VOLTAGE SOURCE CURRENTS 
NAME   CURRENT 
Vs   2.000E+00 
TOTAL POWER DISSIPATION3:  -1.20E+01 WATTS 
SMALL-SIGNAL CHARACTERISTICS 
V(2)/Is = 1.200E+01 
INPUT RESISTANCE AT Is = 1.200E+01 
OUTPUT RESISTANCE AT V(2)= 1.200E+01 
 

L’ equivalente di Thevenin è VTh= V(2) = 30 V e RTh = OUTPUT RESISTANCE = 12 ohm. 

ESEMPIO 13 Determinare l’ampiezza e la fase del fasore di V variando la frequenza da 100Hz a 10kHz in 10 passi. 

0

−

+Vs 10 6 0◦V

1

1kΩ

R1 2

R22kΩ
+
V2

−

C

1µF

 
 
Il file sorgente (ESE_13.CIR ) è scritto di seguito: 
 
ANALISI AC - ESEMPIO 13 
Vs 1 0 AC 10 0 
R1 1 2 1E3 
R2 2 0 2E3 
C 2 0 1E-6 
.AC LIN 10 100 10000 
.PRINT AC Vm(2) Vp(2) 
.PLOT AC Vm(2) Vp(2) 
.PROBE Vm(2) Vp(2) 
.END 

 
Nel file di uscita si ricavano i seguenti risultati: 
 
AC ANALYSIS 
FREQ   VM(2)   VP(2) 
1.000E+02  6.149E+00 -2.273E+01 
1.200E+03  1.301E+00 -7.875E+01 
2.300E+03  6.883E-01 -8.407E+01 
3.400E+03  4.670E-01 -8.598E+01 
4.500E+03  3.532E-01 -8.696E+01 
5.600E+03  2.839E-01 -8.756E+01 
6.700E+03  2.374E-01 -8.796E+01 
7.800E+03  2.039E-01 -8.825E+01 
8.900E+03  1.788E-01 -8.846E+01 
1.000E+04  1.591E-01 -8.863E+01 
 
L’ampiezza e la fase di V sono disegnate di seguito. 

ESEMPIO 14: Determinare l’ equivalente di Thevenin tra i terminali A - B, nel circuito in figura: 
 

1

−

+

7V

VS1

2

6Ω

R4 3

R2

5Ω

3Ω

R1

2Ω
R3

4
−+

VS2

3V
8Ω

R5 5

A

B

0

IS 2.5A

 
 

Anche qui si usa il commando “.TF”. Il file sorgente (ESE_14.CIR ) è scritto di seguito: 
 
ANALISI DC (Equivalente di Thevenin) - ESEMPIO 14 
R1 0 1 3 
R2 0 3 5 
R3 1 3 2 
R4 2 3 6 
R5 4 5 8 
VS1 2 1 DC 7 
VS2 3 4 DC 3 
Is 0 5 DC 2.5 
.TF V(5) VS1 
.END 
 
Nel file di uscita si rilevano i seguenti risultati: 
NODE  VOLTAGE NODE  VOLTAGE  NODE  VOLTAGE 
(1)  3.3947  (2)  10.3950  (3)  6.8421 
(4)  3.8421  (5)  23.8420 
VOLTAGE SOURCE CURRENTS 
NAME   CURRENT 
VS1   -5.921E-01 
VS2   -2.500E+00 
TOTAL POWER DISSIPATION   1.16E+01   WATTS 
V(5)/VS1 = 1.316E-01 
INPUT RESISTANCE AT VS1 = 7.600E+00 
OUTPUT RESISTANCE AT V(5) = 1.037E+01 
 
L’ equivalente di Thevenin è VTh = V5 = 23,8420 V e RTh = 10,37 ohm. 
 
 
 
TRASFORMATORI IDEALI  
 
La sintassi per descrivere nel file sorgente in trasformatore ideale è simile a quella dei induttori mutuamente accoppiati. Il 
primario e il secondario sono degli induttori accoppiati con un coefficiente di accoppiamento pari a 1 (in un trasformatore 
ideale, tutto il flusso generato dall’avvolgimento primario è concatenato con l’avvolgimento secondario). 
La sintassi: 
 

K12 L1 L2 1 
 
descrive un trasformatore K12 con il massimo coefficiente di accoppiamento. 
 
Il rapporto spire con la maggior parte dei trasformatori con il nucleo in Ferro il coefficiente di accoppiamento è 
essenzialmente 1 (spesso o meglio 0.9999999). 
 
I modelli di trasformatori di PSpice usano induttanze (Henry e non spire), pur essendo degli induttori ideali e quindi il 
numero delle spire è proporzionale alla radice quadrata del valore dell’induttanza: per esempio per descrivere un 
trasformatore con rapporto spire pari a 2 (l’avvolgimento secondario ha il doppio delle spire del primario), per cui si 
inserisce un’induttanza sul secondario con valore quattro volte quella del primario. 
 
La sintassi seguente descrive un trasformatore ideale “riduttore” con un rapporto spire n:1 cioè 5:1: 
 

L1 1 0 25 
L2 2 0 1 
K12 L1 L2 1 

 
Entrambi gli avvolgimenti primario e secondario hanno il secondo terminale a massa (nodo 0), perchè PSpice non 
accetta nessun nodo “floating”, cioè senza alcun punto di riferimento. 
 
La polarità del trasformatore consente di ottenere una tensione sul secondario in fase oppure in opposizione di fase 
(180°) rispetto alla tensione sul primario, seguendo il riferimento dei “puntini”. 

ESEMPIO 15: Determinare la corrente nella resistenza RL nel circuito in figura con trasformatore ideale: 

0

−

+VS 5V

1pΩ

R1

L1 L2

3

1kΩ RL

0

1 : 101 2

 
 

 

Il trasformatore ha la tensione secondaria in opposizione di fase rispetto a quella primaria: ciò è evidenziato nella figura 
dai “puntini”, e anche nel file sorgente, dalla sintassi di collegamento dei due induttori (l’induttore L2 del secondario ha 
come primo nodo il nodo 0 di riferimento, mentre per L1 è il contrario). Il generatore di tensione Vs è di tipo sinusoidale di 
5 V con frequenza pari a 1 kHz; il trasformatore è “elevatore” con rapporto spire 10 (1:10), con la tensione secondaria 
sfasata di 180° rispetto a quella primaria. Scopo della simulazione è di verificare le relazioni tra le tensioni e le correnti in 
un trasformatore ideale. Il resistore R1 da 1 p ohm non interferisce sui valori di tensione e corrente ma è “richiesto” da 
PSpice perché nel modello del trasformatore ideale si usano induttori ideali e senza R1 ci sarebbe un cortocircuito. 
 

N.B. PSpice non ammette che un generatore di tensione sia chiuso in cortocircuito (induttori ideali), e che un generatore 
di corrente sia collegato a un circuito aperto (p.e. una maglia di condensatori ideali), perché non riesce a calcolare il 
punto di lavoro in continua (dc steady-state): cosa che fa sempre su tutte le simulazioni che effettua. 
Il file sorgente (ESE_15.CIR ) è scritto di seguito: 
 

Modello di Trasformatore – ESEMPIO 15 
VS 1 0 AC 5 0 
R1 1 2 1E-12 
L1 2 0 1 
L2 0 3 100 
K12 L1 L2 .9999999 
RL 3 0 1E3 
*f=1000/(2*pi)=159.1549 Hz 
.AC LIN 1 159.1549 159.1549 
.PRINT AC I(R1) IP(R1) V(2) VP(2) 
+ I(RL) IP(RL) V(3) VP(3) 
.END 
 

Nel file di uscita si rilevano i seguenti risultati, che confermano quanto affermato precedentemente: 
FREQ  I(R1)  IP(R1)  V(2)  VP(2)  I(RL) 
1.592E+02 5.010E-01 -5.730E-01 5.000E+00 5.733E-14 5.000E-02 
FREQ  IP(RL)  V(3)  VP(3) 
1.592E+02 1.800E+02 5.000E+01 1.800E+02 
 
 
ESEMPIO 16: Determinare le correnti nel primario e secondario nel circuito in figura con trasformatore ideale riduttore: 

0

−

+VS 80 6 45◦V

8Ω

R 2

8mH

L

L1 L2

R2

2Ω

5

125µFC3

0

2 : 11 3 4

 
 

L’analisi è svolta a frequenza “industriale” cioè 50 Hz. Il rapporto di trasformazione è 2 per cui tensione e corrente sul 
secondario saranno rispettivamente la metà e il doppio di quelle sul primario: ciò è dimostrato nella simulazione 
seguente. Il file sorgente (ESE_16.CIR ) è scritto di seguito: 
 

ANALISI AC - Trasformatori ideali - ESEMPIO 16 
Vs 1 0 AC 80 45 
R 1 2 8 
L 2 3 8E-3 
L1 3 0 4 
L2 4 0 1 
K12 L1 L2 .9999999 
R3 4 5 2 
C3 5 0 125E-6 
*f=50 Hz 
.AC LIN 1 50 50 
.PRINT AC I(R) IP(R) I(C3) IP(C3) 
+ V(3) VP(3) V(4) VP(4) 
.END 
 

Nel file di uscita si rilevano i seguenti risultati. Anche qui si verifica quanto affermato precedentemente: 
FREQ  I(R)  IP(R)  I(C3)  IP(C3)  V(3) 
5.000E+01 7.295E-01 1.258E+02 1.588E+00 1.262E+02 8 .111E+01 
FREQ  VP(3)  V(4)  VP(4) 
5.000E+01 4.072E+01 4.055E+01 4.072E+01 

